

Lec3 Simulation with HSpice

School of Micro-Electronics


TJU

Silu Cheng

silu.cheng@tju.edu.cn

Spice Overview

Circuit simulation background



Spice Overview

- **SPICE : Simulation Program with Integrated Circuit Emphasis**
 - Developed by UC Berkeley
- **Numerical Approach to Circuit Simulation**
 - Circuit Node/Connections Define a **Matrix**
- **Must Rely on Sub-Models for Behavior of Various Circuit Elements**
 - Simple (e.g. **Resistor**)
 - Complex (e.g. **MOSFET**)

Spice Overview

- Circuit Analysis tool for Simulation of Electrical Circuits in **Steady-State, Transient, and Frequency Domains**.
- Spice tools available: SBTSPICE, **HSPICE**, Spectre, TSPICE, Pspice, Smartspice ...
- Most of the SPICE tools are originated from **Berkeley's SPICE** program, therefore **support common original SPICE syntax**
- **Basic algorithm** scheme of SPICE tools are similar, however the **control of time step, equation solver** and **convergence control** might be different.

Spice Overview

What you should know before the simulation?

Basic characteristics

Circuit function

Status of the Circuit


Simulation Type

Input signal features

Interdependency and priority among the circuit

General structure and conventions to optimize the circuit

HSPICE Workflow



Supported Analyses

Circuit level simulation and behavior simulation

DC Analysis

AC Analysis

Transient Analysis

Circuit Optimize

Noise Analysis

Fourior Analysis

Monte Carlo Analysis,

Parameter Sweep Analysis


.....

3.2 Input File-Netlist File

Name


the net names each pin is connected to
the value of the device

```
.....  
V1 1 0 10 AC 1  
R1 1 2 1K  
R2 2 0 1K  
C1 2 0 .001U  
.....
```



3.2 Input File-Netlist File

(The Hspice netlist for the RC network circuit) :



```
.title A SIMPLE AC RUN
```

```
.OPTIONS LIST NODE POST
```

```
.OP
```

```
.AC DEC 10 1K 1MEG
```

```
.PRINT AC V(1) V(2) I(R2) I(C1)
```

```
V1 1 0 10 AC 1
```

```
R1 1 2 1K
```

```
R2 2 0 1K
```

```
C1 2 0 .001U
```

```
.END
```

Title line

Optional conditions for simulate

Analyses type

Output file display

Input stimulus

Netlist description

End line

3.2 Input File-Netlist File

Statement or Element	Definition
Title	The first line is the input netlist file title.
* Or \$	Designates comments to describe the circuit
.OPTIONS	Sets conditions for simulation
Analysis statements and .TEMP	Statements to set sweep variables
.PRINT/.PLOT/.GRAPH/.PROBE	Statements to set print, plot and graph variables
.IC or .NODESET	Sets initial state; can also be put in subcircuits
Sources(I or V) and digital to analog inputs	Sets input stimuli
Netlist	Circuit
.LIB	Library
.INCLUDE	General include files
.PROTECT	Turns off output printback
.MODEL libraries	Element model descriptions
.UNPROTECT	Restores output printback
.DELETE LIB	Removes previous library selection
.ALTER	Sequence for in-line case analysis
.END	Required statement to terminate the simulation

3.2 Input File-Netlist File

- No Zip file
- No more than 256 characters.
- Superscript and subscript will be ignored.
- (+) : Continuation marker.
- (*) and (\$) : Common line.
 - * must be the first letter of each line. \$ always follows a complete statement, and has at least one space between \$ and statement.

Hspice的输入——网表文件

- 分隔符包括：tab键，空格，逗号，等号，括号
- 元件的属性由冒号分隔
- 级别由句号指示，例如 X1. A1. V 表示电路X1的子电路A1的节点V
- 常量：
 - **M—毫**，p—皮，n—纳，u—微，**MEG—兆**，例如c1 1 2 10pF;
 - 单位可以省略，例如c1 1 2 10p

电路网表

§ 元件名:

- 元件名以元件的关键字母开头: 电阻—R, 电容—C.....
- 子电路的名字以“X”开头
- 元件名不超过16个字符

§ 节点:

- 节点名长度不超过16个字符, 可以包括句号和扩展名
- 开始的零将被忽略:
- 节点名可以用下列符号开始: # _ ! %
- 节点可以通过 GLOBAL 语句定义成跨越所有子电路的全局节点:

.GLOBAL node1 node2 node3 ...

node1 node2 node3 都是全局节点, 例如电源和时钟名

- 节点0, GND, GND!, GROUND 都指全局的地电位节点

§ 元件语句: 名称 器件的类型 器件所连接的节点 参数值

电路网表

- .TITLE 语句

.TITLE <string of up to 72 characters>

或者: <string of up to 72 characters>

如果是第二种形式, 字符串应该是输入文件的首行;
如果一个HSPICE语句出现在文件的首行, 则它将被认为是标题而不被执行。

- .END 语句

.END <comment>

在 .END语句之后的文本将被当作注释而对模拟没有影响。

3.3 器件定义

无源器件:

- 电阻:

`Rxxx n1 n2 <mname> <R=>resistance <AC=val>`

电阻值可以是表达式。例:

`Rterm input gnd R='sqrt(HERTZ)'`

`Rxxx 9 8 1 AC=1e10` 直流电阻1欧姆, 交流电阻为1e+10欧姆

- 电容:

`Cxxx n1 n2 <mname> <C=>capacitance`

例, `Cload driver output 1.0e-6`。

- 电感:

`Lxxx n1 n2 <L=>inductance`

器件定义

有源器件:

- 二极管:

`DXXX N+ N- MNAME<AREA> <OFF> <IC=VD>`

可选项: *AREA*是面积因子, *OFF*是直流分析所加的初始条件, *IC=VD*是瞬态初始条件模型中的寄生电阻串联在正极端。

- 双极型晶体管:

`QXXX NC NB NE <NS> MNAME<AREA> <OFF> <IC=VBE, VCE>`

NC、*NB*、*NE*、*NS*分别是集电极、基极、发射极和衬底节点, 缺省时*NS*接地。后面与二极管相同。

- JFET:

`JXXX ND NG NS MNAME<AREA> <OFF> <IC=VDS, VGS>`

器件定义

有源器件:

- MOSFET:

`MXXX ND NG NS NB MNAME <L=VAL> <W=VAL> <Other options>`

*M*为元件名称, *ND*、*NG*、*NS*、*NB*分别是漏、栅、源和衬底节点。 *MNAME*是模型名, *L*沟道长, *W*为沟道宽。


- 例: 下面是一个CMOS反相器网表:

.....

M2 out in 0 0 NMOS W=20u L=1u

M1 out in vcc vcc PMOS W=20u L=1u

.....



器件定义

子电路语句:

- 子电路定义开始语句

```
.SUBCKT SUBNAM <node1 node2...> <parnam=val...>
```

其中，SUBNAM为子电路名，node1...为子电路外部节点号，不能为零。子电路中的节点号（除接地点），器件名，模型的说明均是局部量，可以和外部的相同。

例 .SUBCKT OPAMP 1 2 3 4

- 子电路终止语句

```
.ENDS <SUBNAM>
```

若后有子电路名，表示该子电路定义结束；若没有，表示所有子电路定义结束。

例 .ENDS OPAMP

- 子电路调用语句

```
X***** <node1 node2 ...> SUBNAM
```

例 .X1 2 4 17 3 1 MULT1

子电路使用举例

下面是由前面举例的CMOS反相器组成的三级反相器链网表：

.....

```
.global vdd
```

```
.SUBCKT INV IN OUT wn=1.2u wp=1.2u
```

```
Mn out in 0 0 NMOS W=wn L=1.2u
```

```
Mp out in vdd vdd PMOS W=wp L=1.2u
```

```
.ENDS
```

```
X1 IN 1 INV WN=1.2U WP=3U
```


```
X2 1 2 INV WN=1.2U WP=3U
```

```
X3 2 OUT INV WN=1.2U WP=3U
```

```
CL OUT 0 1PF
```

```
VCC VDD 0 5V
```

.....



器件定义

激励源:

•独立源: 电压源—V, 电流源—I

$V_{xxx}/I_{xxx} \quad n^+ \quad n^- \quad \langle \langle DC= \rangle \quad dcval \rangle \quad \langle AC=acmag, \langle acphase \rangle \rangle + \langle M=val \rangle$

例, V1 1 0 DC=5V 或 V1 1 0 5V

I1 1 0 DC=5mA 或 I1 1 0 5mA

交流模式: V1 1 0 AC=10V, 90 幅度为10v, 相位为90度

交直流模式: V1 1 0 0.5v AC=10V, 90 直流分量是0.5v

or

$V_{xxx}/I_{yyy} \quad n^+ \quad n^- \quad \langle tranfun \rangle + \langle M=val \rangle$

tranfun: 瞬态电压源EXP, PULSE, PWL....

$\langle M=val \rangle$ 表示并联的**电流源**个数。

器件定义

激励源:


- 独立源:
- 脉冲形式:

`Vxxx n+ n- PU<LSE> <(>v1 v2 <td <tr <tf <pw <per>>>>> <)>`

V1	值1
V2	值2
td	上升延迟时间
tr	上升时间
tf	下降时间
pw	脉冲宽度
per	周期

脉冲形式举例

例：VPU 3 0 PULSE(1 2 5N 5N 5N 20N 50N)



器件定义

激励源:

- 独立源:
- 正弦形式:

Vxxx n+ n- SIN <(> vo va <freq <td < θ +< ϕ >>>> <)>

v0	失调值
va	幅度
freq	频率
td	延迟时间
θ	阻尼因子
ϕ	相位延迟


器件定义

得到的波形:

Time=0~td	$v_0 + v_a \cdot \sin(2\pi\phi/360 + \text{Time})$
Time=td~瞬态分析的结束时间	$v_0 + v_a \text{Exp}[-(\text{Time} - \text{td}) \times \theta] \cdot \sin\{2\pi \cdot [\text{freq}(\text{Time} - \text{td}) + \phi/360]\}$

正弦形式举例

例: VIN 3 0 SIN (0 1 100MEG 1NS 1e10)



器件定义

激励源：

- 独立源：

- 分段线性形式：

`pwl <(> t1 v1 <t2 v2 t3 v3... > <R <=repeat>>+ <TD=delay> <)>`

v_i 是 t_i 时刻的值，时间点之间线性插值填充，repeat 是开始重复的起始点；delay是延迟时间。

- 指数形式：

`EXP <(> v1 v2 <td1 <t1 <td2 <t2>>>> <)>`

V_1 是初始值， v_2 是峰值， td_1 是上升延迟时间， t_1 是上升时间常数， t_2 是下降时间常数。

完整的网表部分举例


前面反相器链的网表:

.....

```
.SUBCKT INV IN OUT wn=1.2u wp=1.2u  
Mn out in 0 0 NMOS W=wn L=1.2u  
Mp out in vdd vdd PMOS W=wp L=1.2u  
.ENDS
```

```
X1 IN 1      INV WN=1.2U WP=3U  
X2 1  2      INV WN=1.2U WP=3U  
X3 2  OUT    INV WN=1.2U WP=3U  
CL OUT      0      1PF  
VCC  VDD 0 5V  
VIN  IN 0 PULSE(0 5V 10NS 1N 1N 50N 100N)
```

.....



3.4 器件模型

- 模型中列出了一系列元件的类型，并给出了各类型元器件的有关参数，对于不同类型的元件，参数的集合有不同的内容。
- 一个模型对应于一类元件，不同的元件可以对应同一模型，其中各元件间的参数值可能不同，但参数集是一样的，一般值相同的参数的值在模型说明中给出。
- 模型的语句是一条条.MODEL引导的模型说明语句。每个模型有一个名字。

器件模型

无源器件：定义模型参数即可

- 电阻模型 (wire RC) :

.MODEL 模型名 R keyword=value

NOISE, RX: 热噪声参数, $inr = \text{SQRT}(\text{NOISE} \cdot 4KT/R)$,

$$\text{噪声} = inr^2 \cdot RX^2$$

- 电容模型:

.MODEL 模型名 C parameter=value

- 电感模型

.MODEL 模型名 L parameter=value

Capacitance Parameters

Name(Alias)	Units	Default	Description
CAP	F	0	Default capacitance value
CAPSW	F/m	0	Sidewall fringing capacitance
COX	F/m ²	0	Bottomwall capacitance
DEL	m	0	Difference between drawn width and actual width or length DELeff = DEL · SCALM
DI		0	Relative dielectric constant
L	m	0	Default length of capacitor Lscaled = L · SHRINK · SCALM
SHRINK		1	Shrink factor
TC1	1/deg	0	First temperature coefficient for capacitance
TC2	1/deg ²	0	Second temperature coefficient for capacitance
THICK	m	0	Insulator thickness
TREF	deg C	TNOM	Reference temperature
W	m	0	Default width of capacitor Wscaled = W · SHRINK · SCALM

器件模型

有源器件的模型说明都有一个LEVEL参数，不同的LEVEL对应不同的精度。

- 二极管模型：包括齐纳二极管、Schottky、硅扩散结

`.MODEL 模型名 D <LEVEL = val> <keyword = val> ...`

- nongeometric junction diode: LEVEL=1

电阻、电容、电流参数

例： `.MODEL D D (CO=2PF, RS=1, IS=1P)`

`.MODEL DFOWLER D (LEVEL=2, TOX=100, JF=1E-10, EF=1E8)`

`.MODEL DGEO D (LEVEL=3, JS=1E-4, JSW=1E-8)`

- geometric junction diode: 芯片中的二极管 (LEVEL=3)

金属、多晶层的几何参数

模型说明中涉及的参数需与 .OPTIONS 的设置相配合：

器件模型

- BJT模型:

.MODEL mname NPN <(> <pname1 = val1> ...<)>

or

.MODEL mname PNP <pname1 = val1> ...

模型参数中一般包括LEVEL，说明哪种模型，不同级的模型有不同的模型参数集。

器件模型

- MOS模型:

- .MODEL 模型名 PMOS <LEVEL=val> <parameters>

- .MODEL 模型名 NMOS <LEVEL=val> <parameters>

- LEVEL=1 常用于数字电路, 精度低、速度快

- LEVEL=2 耗尽型MOSFET

- LEVEL=13, 39, 49模拟电路, 精度高、速度慢

MOSFET模型说明举例

例：1.2um CMOS工艺MOS管SPICE模型：

```
.MODEL NMOS NMOS LEVEL=2 LD=0.15U TOX=200.0E-10 VTO=0.74 KP=8.0E-05  
+NSUB=5.37E+15 GAMMA=0.54 PHI=0.6 U0=656 UEXP=0.157 UCRIT=31444  
+DELTA=2.34 VMAX=55261 XJ=0.25U LAMBDA=0.037 NFS=1E+12 NEFF=1.001  
+NSS=1E+11 TPG=1.0 RSH=70.00 PB=0.58  
+CGDO=4.3E-10 CGSO=4.3E-10 CJ=0.0003 MJ=0.66 CJSW=8.0E-10 MJSW=0.24
```

```
.MODEL PMOS PMOS LEVEL=2 LD=0.15U TOX=200.0E-10 VTO=-0.74 KP=2.7E-05  
+NSUB=4.33E+15 GAMMA=0.58 PHI=0.6 U0=262 UEXP=0.324 UCRIT=65720  
+DELTA=1.79 VMAX=25694 XJ=0.25U LAMBDA=0.061 NFS=1E+12 NEFF=1.001  
+NSS=1E+11 TPG=-1.0 RSH=121.00 PB=0.64  
+CGDO=4.3E-10 CGSO=4.3E-10 CJ=0.0005 MJ=0.51 CJSW=1.35E-10 MJSW=0.24
```

直流分析仿真流程




Figure 9-1: DC Initialization and Operating Point Analysis Simulation Flow

3.5 分析类型描述语句

• .OP: 工作点分析

会在输出文件中列出一些直流参数和各结点的工作点电压与支路电流、静态功耗。

- 一般在任何其它仿真之前都需要计算直流工作点。
- .op time 看某一时刻的各个器件和电源的状态。
- .op vol time 某一时刻所有节点电压。
- .op cur time 某一时刻所有器件和电源的电流。
- 默认状态: .ac和.dc分析是初始工作点, .tran默认时间是0。

分析类型描述语句

- .DC: 直流扫描分析
 - 在直流分析中，.DC 语句可进行
 - a. 直流参数值扫描
 - b. 电源值扫描
 - c. 温度范围扫描
 - d. 执行直流蒙特卡罗分析（随机扫描）
 - e. 完成直流电路优化
 - f. 完成直流模型特性化

分析类型描述语句

- .DC 语句具体格式取决于实际应用需要，下面给出了一些应用格式：

(1) 直流扫描或直流参数化扫描

`.DC var1 start1 stop1 step1/<SWEEP var2 type np start2 stop2>`

- 注意：

1) 对独立源扫描时，var为电源名，而非节点名

2) 仿真迟滞特性要正、负双向扫描，

如： `.DC Vin 0 5 0.1`

`.DC Vin 5 0 -0.1`

3) 注意对两个量的扫描，后面的是外循环

4) SWEEP后的变量可是电压、电流或温度等变量

- 例： `.DC xval 1k 10k .5k SWEEP TEMP LIN 5 25 125`

`.DC TEMP POI 5 0 30 50 100 125`

对前面反相器链的直流特性扫描：

`VIN IN 0`

`.DC VIN 0 5V 0.1V (从0v到5v，步长0.1v)`

直流分析举例

例：分析反相器链的直流传输特性和工作点

.....

```
.global vdd
```

```
.SUBCKT INV IN OUT wn=1.2u wp=1.2u
```

.....

```
.ENDS
```

```
X1      IN      1      INV      WN=1.2U WP=3U
```

```
X2      1        2      INV      WN=1.2U WP=3U
```

```
X3      2        OUT    INV      WN=1.2U WP=3U
```

```
CL      OUT      0      1PF
```

```
VCC  VDD 0 5V
```


```
VIN  IN   0
```

```
.DC VIN      0 5V 0.1V/SWEEP TEMP -20 80 1
```


```
.OP
```

.....

```
.END
```



在Metawave中输出的直流传输特性曲线



瞬态分析仿真流程




Figure 10-1: Transient Analysis Simulation Flow

分析类型描述语句——瞬态分析

- 瞬态分析是与时间有关的电路特性分析。HSPICE 在进行瞬态分析时首先执行直流工作点分析，并以此为初始条件开始进行瞬态分析，除非在.TRAN 语句中也包含了UIC 参数，则.TRAN 语句进行瞬态分析时，从.IC 语句中规定的节点电压开始进行。
- 值得注意的是，对于振荡器电路或带有反馈的电路，因为其没有稳态工作点，所以在瞬态分析时，必须把反馈回路打断，这样瞬态分析开始前才能进行直流工作点计算。

分析类型描述语句——瞬态分析

- 一般分析:

`.TRAN var1 START=start1 STOP=stop1 STEP=incr1`

or

`.TRAN tincr1 tstop1 <tincr2 tstop2 ... tincrN tstopN>`

`+ <START=val> <UIC>`

起始时刻和步长都指的是输出打印的时刻点，计算的时间步长由hspice自己决定。UIC参数表示使用.IC语句指定的节点初始值。

例: `.TRAN .1NS 25NS 1NS 40NS START=10NS`

0—25ns, 步长0.1ns, 25ns—40ns, 步长1ns; 从10ns开始输出结果。

分析类型描述语句——瞬态分析

- 傅立叶分析是瞬态分析的一部分，HSPICE 在 $t_{stop} - f_{period}$ 到 t_{stop} 的时间间隔内进行傅立叶分析， t_{stop} 就是 .TRAN 语句中的终止时间，而 f_{period} 是傅立叶分析的基频。
- 傅立叶分析时不是所有的瞬态结果都要用到，只用到瞬态分析分析终止时间 T_{STOP} 之前的基频的一个周期。傅立叶分析时间是 $T_{STOP} - (T_{STOP} - 1/f_{period}) = 1/f_{period}$ ，就是说，瞬态分析至少要坚持 $1 / f_{period}$ 。傅立叶分析能够得到基频、DC 分量和第2 到第9 次谐波（交流分量）。

分析类型描述语句——瞬态分析

•Fourier分析:

`.FOUR freq ov1 <ov2 ov3 ...>`

Freq—基频, ov1、ov2 . . . —输出变量

例:

CMOS INVERTER

M1 2 1 0 0 NMOS W=20U L=5U

M2 2 1 3 3 PMOS W=40U L=5U

VDD 3 0 5

VIN 1 0 SIN 2.5 2.5 20MEG

.MODEL NMOS NMOS LEVEL=3 CGDO=.2N CGSO=.2N CGBO=2N

.MODEL PMOS PMOS LEVEL=3 CGDO=.2N CGSO=.2N CGBO=2N

.OP

.TRAN 1N 100N

.FOUR 20MEG V(2)

.PRINT TRAN V(2) V(1)

.END

输出文件傅立叶分析结果部分:

fourier components of transient response v(2)

dc component = 2.430D+00

harmonic frequency fourier

		normalized		phase	normalized	
no	(hz)	component	component	(deg)	phase(deg)	
1	20.0000x	3.0462	1.0000	176.5386	0.	
2	40.0000x	115.7006m	37.9817m	-106.2672	-282.8057	
3	60.0000x	753.0446m	247.2061m	170.7288	-5.8098	
4	80.0000x	77.8910m	25.5697m	-125.9511	-302.4897	
5	100.0000x	296.5549m	97.3517m	164.5430	-11.9956	
6	120.0000x	50.0994m	16.4464m	-148.1115	324.6501	
7	140.0000x	125.2127m	41.1043m	157.7399	-18.7987	
8	160.0000x	25.6916m	8.4339m	172.9579	-3.5807	
9	180.0000x	47.7347m	15.6701m	154.1858	-22.3528	

total harmonic distortion = 27.3791 percent

分析类型描述语句——交流分析

- .AC:

- 一般频域扫描:

.AC type np fstart fstop <SWEEP var start stop incr>

or

.AC type np fstart fstop <SWEEP var type np start stop>

or

.AC var1 START = start1 STOP = stop1 STEP = incr1

例: .AC DEC 10 1K 100MEG 1kHz-100MHz,每10倍频10个采样点。

DEC –十进制的(decade variation)

OCT –八进制的(octave variation)

LIN –线形的(linear variation)

POI –列举的(list of points)

分析类型描述语句—交流分析

- 噪声分析:

用来计算各个器件的噪声对输出节点的影响并给出其均方根并输出, 可完成. AC语句规定的各频率的计算, 应在. AC分析之后。

`.NOISE ovv srcnam inter`

ovv—输出变量, srcnam—输入源, inter—频率间隔

例: `.title ac sweep example`

`.OPTIONS POST`

`R1 in 1 5`

`C1 1 0 500pf`

`V1 IN 0 0 AC=10V,37`

`.AC OCT 10 1 100MEG`

`.noise v(1) v1 20` —分析1点电压的噪声情况, 噪声源为V1端口

`.END`

Lis文件中输出的噪声分析结果

```
1 ***** Star-HSPICE -- 1999.4 (19991220) 22:12:12 04/16/2002 pcnt
```

```
*****
```

```
.title ac sweep example
```

```
***** noise analysis          tnom= 25.000 temp= 25.000
```

```
*****
```

```
frequency = 1.0000 hz
```

```
**** resistor squared noise voltages (sq v/hz)
```

```
element 0:r1
```

```
total 8.233e-20
```

```
rx 5.0000
```

```
**** total output noise voltage      = 8.233e-20 sq v/hz
```

```
      = 286.9260p v/rt hz
```

```
transfer function value:
```

```
v(1)/v1          = 1.0000
```

```
equivalent input noise at v1      = 286.9260p /rt hz
```

```
**** the results of the sqrt of integral (v**2 / freq)
```

```
from fstart upto 1.0000 hz. using more freq points
```

```
results in more accurate total noise values.
```

```
**** total output noise voltage = 0. volts
```

```
**** total equivalent input noise = 0.
```

```
.....
```

最后给出总的输出噪声电压和等效到输入端的噪声电压

参数扫描

1. 要定义扫描的参数;
2. 在电路中引用参数;
3. 给出参数取值列表 (DATA);
4. 在分析语句中加入DATA=datanam, 指定参数对应的数值表。

参数扫描举例

```
.title ac sweep example
```

```
.OPTIONS POST
```

```
.param cv=500pf
```

```
R1 in 1 5
```

```
C1 1 0 cv
```

```
V1 IN 0 0 AC=10V,37
```

```
.data cv_table
```

```
cv
```

```
300p
```

```
500p
```


```
1n
```

```
.enddata
```

```
.AC OCT 10 1 100MEG sweep data=cv_table
```

```
*.PRINT ac V(1)
```

```
.END
```



分析类型描述语句

•温度分析：

与直流或瞬态分析等命令结合使用：例如对反相器链瞬态特性的温度扫描：

.....

```
VIN      IN 0 PULSE(0 5V 10NS 1N 1N 50N 100N)
```


```
.TRAN 1N 200N sweep temp 0 125 20
```

```
.PRINT V(OUT)
```

```
.END
```

曲线如右：

.TEMP t1 <t2 <t3 ...>>：会产生一系列的瞬态分析文件：tr0, tr1..., 在metawave中对应不同的分析。



分析类型描述语句

初始化:

- 瞬态分析的初始化语句

`.IC var1=val1 <var2=VAL2> . . .`

- 节点电压初始条件设定语句

`.NODESET var1=val1 <var2=VAL2> . . .`

进行含有多稳态电路的模拟时往往需要给出初始化条件。

例：含有双稳态的电路

固定A端为1，扫描输出一输入B的直流特性：

`va a 0 5v`

`vb b 0`

`.dc vb lin 10 0 5`

不收敛！应加上

`.nodeset v(i)=5v v(out)=0v`

瞬态分析需要初始化：

...


`va a 0 5v`

`vb b 0 pulse(0 5 1n 0.1n 0.1n 5n 10n)`

`.IC v(i)=5v v(out)=0v`

`.tran 0.1n 20ns`

...



3.6 控制语句

- 控制语句是hspice输入文件的命令部分，告诉hspice要进行哪些操作和运算，并给出相关的参数——如分析方式、输出的变量等。
- 其内容主要包括选项语句（.OPTIONS）、分析命令语句、输出控制语句几类。这些语句格式的共同特点是都由保留字引导，后面跟随相应的参数，在保留字前要加“.”

控制语句

- .INCLUDE语句: 引用一个文件

例:

```
.include '<filepath>filename'
```

- LIB 语句:

```
.lib '<filepath>filename' entryname
```

该语句根据文件路径和文件名来调用一个库文件，一般该文件包含常用命令、器件模型、子电路分析及定义。

例 .lib 'MODELS' cmos1

MODELS文件:

...

```
.MODEL CMOS1 nmos ...
```

控制语句

- .PARAM 语句

定义一个变量

`.PARAM <parametername> = ' <Expression> '`

例: `.PARAM width = 20u`

`.PARAM length = 'sqrt(width)*1.65'`

`M1 3 2 0 0 NMOS width length`

- .DATA 语句:

给一个数组赋值。

例: `.DATA D1 width length RL`

`+           50u     20u     1K`

`+           60u     10u     10K`

`+          100u     25u     1K`

`.ENDDATA`

控制语句

- . GLOBAL 语句

定义全局变量

`.GLOBAL parametername (XX)`

例：.GLOBAL VDD

.GLOBAL VSS

- 注：上述语句会把所有命名为XX的节点连接起来，包括子电路里面的节点。
- Global的使用需格外小心，一般仅把电源定义成Global

控制语句

- **.OPTIONS:**

该语句允许用户重新设置程序的参数或控制程序的功能。常用的一些如下：

node: 列出个节点的元件端点，便于查错；

post: 使输出数据可以使用 AvanWaves 浏览(即将数据输出到post processor) ；

list: 列出元件列表；

MEASDGT: .MEASURE语句输出的有效数字位数

例: .option post probe

只观察.probe语句输出的变量。

3.7 输出控制

- 输出语句:

- **.PRINT**: 在输出的list文件中打印数字的分析结果, 如果.OPTIONS中有POST则同时输出到post-processor中。
- **.PLOT**: 在输出的list文件中打印低分辨率的曲线 (由ASCII字符组成), 如果.OPTIONS中有POST则同时输出到post-processor中。
- **.GRAPH**: 生成用于打印机或PostScript格式的高分辨率曲线。
- **.PROBE**: 把数据输出到post-processor, 而不输出到list文件。
- **.MEASURE**: 输出用户定义的分析结果到mt0文件, 如果.OPTIONS中有POST则同时输出到post-processor中。
- **.OP**, **.TF**, **.NOISE**, **.SENS**和**.FOUR**都提供直接输出功能。

输出控制

- **.PRINT:**

.PRINT antype ov1 <ov2 ... ov32>

Antype – AC/DC/TRAN;

Ovi: 输出变量，可以有以下形式:

V (1) 节点1的电平, v (1, 2) 1、2间的电压, V (R1) 电阻R1的电压;

VM (1) v1的幅值, VR (1) v1的实部, VI (1) v1的虚部, VP (1) v1的相位, VDB (1) v1的分贝值;

(电流与以上类似) ;

输出控制

- **.PLOT**:在输出列表文件中输出低分辨率的点

`.PLOT antype ov1 <(plo1,phi1)> ... <ov32>
+<(plo32,phi32)>`

(plo1,phi1)—ov1绘图的上下限。

- **.PROBE**:输出图形数据但不在输出列表文件中输出

`.PROBE antype ov1 ... <ov32>`

元件电流引用:

BJT: I1(Qx)-Ic, I2 (Qx)-Ib, I3(Qx)-Ie, I4(Qx)-衬底电流;

MOS:I1(Mx)-Id, I2(Mx)-Ig, I3(Mx)-Is, I4(Mx)-衬底电流。

输出控制

- **.MEASURE:**

包括以下测量模式:

- Rise, fall, and delay
- Find-when
- Equation evaluation
- Average, RMS, min, max, and peak-to-peak
- Integral evaluation
- Derivative evaluation
- Relative error

- **.MEASURE <DC|AC|TRAN> result TRIG ... TARG ...**

Result是测量结果的名字, TRIG ... TARG表示起始...中止, 对瞬态分析它是时间, 对AC分析它是频率, 对DC分析它是DC扫描变量。如果target在trigger激活以前到达, 结果的值是负的。

输出控制

.MEASURE:

- Rise, Fall, Delay模式:

.MEASURE <DC|AC|TRAN> result TRIG ... TARG ...

Result—测量结果的名字, TRIG ... TARG 一起始…中止 (依分析内容不同可是时刻、频率…)

- TRIG和TARG的格式

TRIG trig_var VAL=trig_val <TD=time_delay> <CROSS=c>

+ <RISE=r> <FALL=f> 或 TRIG AT=val

TARG targ_var VAL=targ_val <TD=time_delay>

+ <CROSS=c|LAST> <RISE=r|LAST> <FALL=f|LAST>

trig_var和targ_var指定引发变量; val指出上升、下降、或反转的临界点; time_delay指出开始测量时跳过的时间量; CROSS, RISE, FALL分别指出开始触发的次数; LAST说明到最后一次;

例 .meas tran tdlay trig v(1) val=2.5 td=10n rise=2

+ targ v(2) val=2.5 fall=2

计算反相器链电路的延迟时间

```
.TITLE 1.2UM CMOS INVERTER CHAIN
.INCLUDE "models.sp"
.....
X1 IN 1      INV WN=1.2U WP=3U
X2 1  2      INV WN=1.2U WP=3U
X3 2  OUT    INV WN=1.2U WP=3U
CL OUT      0      1PF
VCC VDD 0 5V
VIN  IN 0 PULSE(0 5V 10NS 1N 1N 50N 100N)
.TRAN 1N 200N
.measure tran tdelay trig v(in) val=2.5 td=8ns rise=1
+      targ v(out) val=2.5 td=9n fall=1
.END
```

测量结果

输出的Chain.mt0文件:


```
$DATA1 SOURCE='HSPICE'  
VERSION='1999.4'
```

```
.TITLE '.title 1.2um cmos  
inverter chain'
```

```
tdelay temper alter#
```

```
9.121e-09 25.0000 1.0000
```

延迟**9.121ns**



Hspice的输出

•.MEASURE:

Find—When模式:

.MEASURE <DC|TRAN| AC> result 条件 < RISE=r | LAST >
+< FALL=f | LAST>< CROSS=c | LAST >

条件: WHEN out_var = val或WHEN out_var1=out_var2或 FIND
out_var1 WHEN out_var2=val或FIND out_var1 WHEN out_var2 =
out_var3

例: 若在控制卡中加如下一句:

.measure tran ttrans when v(out)=4.5v

求输出电压降到4.5v的时刻, 则输出文件chain.mt0如下:

```
$DATA1 SOURCE='HSPICE' VERSION='1999.4'  
.TITLE '.title 1.2um cmos inverter chain'  
tdelay ttrans temper alter#  
9.121e-09 1.262e-08 25.0000 1.0000
```

分析功耗

- 分析功耗

`.print/plot <dc或tran> P(element) power`

其中，power 关键词用来计算整个电路的功率。

例 `.print tran P(M1) P(Vin) P(Cload) Power`

*这里只计算瞬态分析或直流分析中的瞬时功耗或静态功耗。

- 分析效率


`.meas tran pload avg p(IRL) from=700u to=800u`

这个语句是测量700u~800u这一段时间的效率，IRL是输出负载电流，AVG是求平均，效率是负载上的功耗除以输入功耗。

3.7 使用 **Awaves** 浏览波形 ——HSPICE分析结果的浏览

Hspice Graphic Tools: **Awaves**

Awaves - Awaves Window




Hspice Graphic Tools: Awaves

Awaves - Awaves Window

menu


Tool button

waveform




Data names

Awaves- Open Design




Awaves- Result Browser



Awaves - Measurement

On window measurement function




Awaves - Multiple Panels

Multiple panels can be displayed on window


Maximum number of panels displayed depends on window size

Waveforms can be dragged and drop into other panels




Awaves - View Port Management

Zoom size is independent between panels



Awaves - Expressions

Expressions provide capability for data calculation
Can use drag&drop to type expression



一个完整例子：缓冲驱动器分析

——HSPICE分析举例

准备模型文件

- 选用1.2um CMOS工艺level II模型(Models.sp)

```
.MODEL NMOS NMOS LEVEL=2 LD=0.15U TOX=200.0E-10 VTO=0.74 KP=8.0E-05  
+NSUB=5.37E+15 GAMMA=0.54 PHI=0.6 U0=656 UEXP=0.157 UCRIT=31444  
+DELTA=2.34 VMAX=55261 XJ=0.25U LAMBDA=0.037 NFS=1E+12 NEFF=1.001  
+NSS=1E+11 TPG=1.0 RSH=70.00 PB=0.58  
+CGDO=4.3E-10 CGSO=4.3E-10 CJ=0.0003 MJ=0.66 CJSW=8.0E-10 MJSW=0.24
```

```
.MODEL PMOS PMOS LEVEL=2 LD=0.15U TOX=200.0E-10 VTO=-0.74 KP=2.70E-05  
+NSUB=4.33E+15 GAMMA=0.58 PHI=0.6 U0=262 UEXP=0.324 UCRIT=65720  
+DELTA=1.79 VMAX=25694 XJ=0.25U LAMBDA=0.061 NFS=1E+12 NEFF=1.001  
+NSS=1E+11 TPG=-1.0 RSH=121.00 PB=0.64  
+CGDO=4.3E-10 CGSO=4.3E-10 CJ=0.0005 MJ=0.51 CJSW=1.35E-10 MJSW=0.24
```


设计基本反相器单元

- 根据模型参数、设计要求设定管子尺寸
- 写出反相器网单:

```
.TITLE 1.2UM CMOS INVERTER CHAIN
```

```
.INCLUDE "models.sp"
```

```
.global vdd
```

```
Mn out in 0 0 NMOS W=1.2u L=1.2u
```

```
Mp out in vdd vdd PMOS W=3u L=1.2u
```

```
CL OUT 0 0.5PF
```

```
VCC VDD 0 5V
```

```
VIN IN 0 PULSE(0 5V 10NS 1N 1N 50N 100N)
```

```
.....
```


直流传输特性分析

```
.TITLE 1.2UM CMOS INVERTER CHAIN
.INCLUDE "models.sp"
.global vdd
.option probe
Mn out in 0 0 NMOS W=1.2u L=1.2u
Mp out in vdd vdd PMOS W=1.2u L=1.2u
CL OUT      0      0.5PF
VCC  VDD 0 5V
VIN   IN 0 PULSE(0 5V 10NS 1N 1N 50N 100N)


.DC VIN      0 5V 0.1V
.op
.probe v(out)
.end
```

- 启动hspice
- 在Edit NL输入网表
- 点击Simulate
- 在Edit LL看输出文件

启动awave 选择Design->open, 在open design窗口中从Filter菜单中选中input或all, 会在下面文件列表中显示出来, 任选一个设计文件。




Avanwave中输出的直流传输特性曲线



利用含参数的子电路组成反相器链

```
.TITLE 1.2UM CMOS INVERTER CHAIN
.INCLUDE "models.sp"
.global vdd
.SUBCKT INV IN OUT wn=1.2u wp=1.2u
Mn out in 0 0 NMOS W=wn L=1.2u
Mp out in vdd vdd PMOS W=wp L=1.2u
.ENDS
```

```
X1 IN 1      INV WN=1.2U WP=3U
X2 1  2      INV WN=1.2U WP=3U
X3 2  OUT    INV WN=1.2U WP=3U
CL OUT      0      1PF
VCC  VDD 0 5V
VIN  IN 0
.....
```



直流特性分析

.....

```
.DC VIN 0 5V 0.1V
```

```
.measure DC tran ttrans  
when v(out)=2.5v
```


.....

```
.END
```

.....

```
ttrans temper alter#
```

```
2.4500 25.0000 1.0000
```



时序特性

.....

```
VIN  IN 0 PULSE(0 5V 10NS  
1N +1N 50N 100N)
```

```
*.DC VIN  0 5V 0.1V
```


```
.TRAN 1N 200N
```

```
.measure tran tdelay trig v(in)  
val=2.5 td=8ns rise=1
```

```
+      targ v(out) val=2.5 td=9n  
fall=1
```

```
.PRINT V(OUT)
```


```
.end
```



考察驱动能力

- 扫描负载电容，观察时序波形：

```
.....  
.param cload=1pf  
.....  
.data cv  
cload  
0.5p  
1p  
2p  
.enddata  
.....  
CL          OUT 0 cload  
VIN         IN 0 PULSE(0 5V 10NS 1N  
            1N  
            +50N 100N)  
.TRAN 1N 200N sweep data=cv  
.....
```



固定负载，扫描管子尺寸

.....

.param cload=1.2u

.param wpt='2.5*clload'

.....

.data cv

clload

1.2u 2.4u 3u

.enddata

X1 IN 1 INV WN=clload WP=wpt

X2 1 2 INV WN=clload WP=wpt

X3 2 OUT INV WN=clload WP=wpt

CL OUT0 1pf

.....

.TRAN 1N 200N sweep data=cv

.measure tran td trig v(in) val=2.5 td=8ns

+ rise=1 targ v(out) val=2.5 td=9n fall=1

.END

\$DATA1 SOURCE='HSPICE' VERSION='1999.4'

.TITLE '.title 1.2um cmos inverter chain'

index clload td temper alter#

1.0000 1.200e-06 9.121e-09 25.0000

1.0000

2.0000 2.400e-06 4.724e-09 25.0000

1.0000

3.0000 3.000e-06 3.891e-09 25.0000

1.0000

优化的各级反相器尺寸

```
.....
X1 IN 1      INV WN=1.2u WP=3u
X2 1  2  INV WN=2.4u WP=6u
X3 2  OUT INV WN=4.8u WP=12u
CL OUT 0    1pf
.....
.TRAN 1N 200N $sweep data=cv
.measure tran tdelay trig v(in) val=2.5 td=8ns
+rise=1 targ v(out) val=2.5 td=9n fall=1
.....
面积和延迟（他的tdelay）都比前面3个wn全为
3u的情况小。
```

```
$DATA1 SOURCE='HSPICE'
VERSION='1999.4'
.TITLE '.title 1.2um cmos
inverter chain'
tdelay temper alter#
3.011e-09 25.0000
1.0000
```

温度特性扫描

.....
.TRAN 1N 200N sweep temp -
+10 125 10 从10 °C到125 °C
.....

